

## SPIKE BASE

### PURPOSE

Specifies the base location for spikes on plots.

### DESCRIPTION

A spike is a vertical line from the base number to the plot point. Spike bases are specified by trace and are given in terms of the plot units.

### SYNTAX

SPIKE BASES <number> <number> <number> etc.

where <number> is a number or parameter that specifies the desired spike base. Up to 100 spike bases can be specified.

### EXAMPLES

```
SPIKE BASES 0. 0. 10.  
SPIKE BASES 20. 20. 20.  
SPIKE BASES 0. ALL  
SPIKE BASES ALL 0.  
SPIKE BASES
```

### NOTE

The SPIKE BASE command with no arguments sets the spike base to 0.0 for all traces. The SPIKE BASE command with the word ALL before or after the specified base assigns that spike base to all traces; thus SPIKE BASE 0.0 ALL or SPIKE BASE ALL 0.0 uses a base of 0.0 for all traces.

### DEFAULT

All spike bases are 0.0.

### SYNONYMS

None

### RELATED COMMANDS

PLOT	=	Generates a data or function plot.
SPIKE	=	Sets the on/off switches for plot spikes.
SPIKE COLOR	=	Sets the colors for plot spikes.
SPIKE DIRECTION	=	Sets the directions for plot spikes.
SPIKE LINE	=	Sets the line types for plot spikes.
SPIKE THICKNESS	=	Sets the line thicknesses for plot spikes.

### REFERENCES

“Elements of Graphing Data,” William S. Cleveland, Wadsworth Advanced Books and Software, 1985.

“Visualizing Data,” William S. Cleveland, Hobart Press, 1993.

### APPLICATIONS

Presentation graphics, time series plots, dot charts

### IMPLEMENTATION DATE

Pre-1987

## PROGRAM

```

SKIP 25
READ ELNINO.DAT Y YEAR MONTH
RETAIN Y YEAR MONTH SUBSET YEAR > 1977
SKIP 0
.
LET JUNK = DISTINCT YEAR; LET NYRS = SIZE JUNK
LET JUNK2 = DISTINCT MONTH; LET NGROUP = SIZE JUNK2
LET NSPACE = 2; LET NTEMP = NYRS+NSPACE
LET NSTOP = NTEMP*NGROUP; LET X = SEQUENCE 1 NTEMP NSTOP
FEEDBACK OFF
LOOP FOR K = 2 1 NYRS
    LET TEMP = SEQUENCE K NTEMP NSTOP
    EXTEND X TEMP
END OF LOOP
.
MULTILOT 2 1; MULTILOT CORNER COORDINATES 0 0 100 100
FRAME CORNER COORDINATES 15 20 95 95
SPIKE ON; LINE BLANK
LET A = MEAN Y
SPIKE BASE A
X1LABEL MEAN PLOT; Y1LABEL SOUTHERN OSCILLATION INDEX
XLIMITS 1 NGROUP; XTIC OFFSET 1 1
MAJOR XTIC MARK NUMBER NGROUP; MINOR XTIC MARK NUMBER 0
X1TIC MARK LABEL FORMAT ALPHA
X1TIC MARK LABEL CONTENTS JAN FEB MARCH APRIL MAY JUNE JULY AUG SEP ...
    OCT NOV DEC
MEAN PLOT Y MONTH
READ STRING SP
.
LET A = YPLOT(1)
LET STRING S = ^A&^SP
LET XL = DATA 1 NYRS; LET YL = DATA A A; LET TAGL = DATA 1 1
LET I2 = 2
LOOP FOR K = 2 1 NGROUP
    LET A = YPLOT(K); LET A = INT(100*A)/100
    LET STRING T = ^A; LET STRING S = ^S&^SP&^T
    LET I1 = I2 + 1; LET I2 = I1 + 1
    LET YL = A FOR I = I1 1 I2; LET TAGL = K FOR I = I1 1 I2
    LET A1 = (K-1)*NTEMP+1; LET A2=A1+NYRS-1
    LET XL = DATA A1 A2 FOR I = I1 1 I2
END OF LOOP
SPIKE BASE ^S
.
LINE BL BL BL BL BL BL BL BL BL BL BL BL BL BL BL BL
SPIKE ON ON ON ON ON ON ON ON ON ON ON ON ON ON ON ON
LET A = 1 + NYRS; XTIC OFFSET 1 A
LET A = NSTOP - NYRS
XLIMITS 1 A
XTIC MARK LABEL JUSTIFICATION LETO
X1LABEL SEASONAL SUBSERIES PLOT
YLIMITS -3.5 1.5
YTIC OFFSET 0.15 0.15
PLOT Y X MONTH AND
PLOT YL XL TAGL
END OF MULTILOT

```

